

```
1 ' -----
2 '     From the book Inside Active Directory, ISBN 0-201-61621-1
3 '     Copyright (C) 2002 by Addison-Wesley
4 '     Script by Sakari Kouti (see http://www.kouti.com)
5 ' You have a royalty-free right to use, modify, reproduce and distribute
6 ' this script (and/or any modified version) in any way you find useful,
7 ' provided that you agree that Addison-Wesley or Sakari Kouti has no
8 ' warranty, obligations or liability for the script. If you modify
9 ' the script, you must retain this copyright notice.
10 ' -----
11 Option Explicit
12
13 Const UF_SMARTCARD_REQUIRED = &H40000
14
15 Call CreateUser("Jack", "Brown")
16
17 Sub CreateUser(strFirstName, strLastName)
18     Dim objContainer, objUser, objLargeInt, strSAMName, intUserFlags
19
20     Set objContainer = _
21         GetObject("LDAP://OU=Sales,DC=sanao,DC=com")
22
23     Set objUser = objContainer.Create("user", _
24         "CN=" & strLastName & " " & strFirstName)
25
26     'The Essential Properties
27     '-----
28     strSAMName = strFirstName & Left(strLastName, 1)
29     Call objUser.Put("sAMAccountName", strSAMName)
30     Call objUser.Put("userAccountControl", &H200)
31     Call objUser.Put("userPrincipalName", _
32         strFirstName & "." & strLastName & "@sanao.com")
33
34     objUser.SetInfo
35     objUser.SetPassword ("secret")
36     objUser.GetInfo
37
38     'The Significant Properties
39     '-----
40     Call objUser.Put("profilePath", "\\dc1\Prof$\\" & strSAMName)
41     'We let Group Policy take care of the logon script setting
42     Call objUser.Put("homeDirectory", "\\dc1\" & strSAMName)
43     Call objUser.Put("homeDrive", "H:")
44
45     'User must NOT change password at next logon
46     Set objLargeInt = CreateObject("LargeInteger")
47     objLargeInt.LowPart = &HFFFFFFF
48     objLargeInt.HighPart = &HFFFFFFF
49     Call objUser.Put("pwdLastSet", objLargeInt)
50
51     objUser.AccountDisabled = True
52
53     intUserFlags = objUser.Get("userAccountControl")
54     intUserFlags = intUserFlags Or UF_SMARTCARD_REQUIRED
55     'If we wanted to clear a setting instead of setting it:
56     'intUserFlags = intUserFlags And Not UF_SMARTCARD_REQUIRED
57     Call objUser.Put("userAccountControl", intUserFlags)
58
59     'The Informational Properties
60     '-----
61     Call objUser.Put("givenName", strFirstName)
62     Call objUser.Put("initials", _
63         Left(strFirstName, 1) & Left(strLastName, 1))
64     Call objUser.Put("sn", strLastName)
65     Call objUser.Put("displayName", strFirstName & " " & strLastName)
66     Call objUser.Put("telephoneNumber", "555-4268")
67     Call objUser.Put("otherTelephone", Array("555-6122", "555-7991"))
68
```

```
69     objUser.SetInfo
70     Set objUser = Nothing
71 End Sub
72
```